VA for internal NLS soliton oscillations using a Gaussian ansatz

```
> restart;
> interface(showassumed=0):;
```

Ansatz of the form u(x,t) = A(x,t) * exp(I*phi(x,t))

```
> A := B(t)*exp(-(x)^2/(2*w(t)^2));
> phi := d(t)*(x)^2+b(t);
```

$$A := B(t)\, \mathrm{e}^{-\frac{x^2}{2\, w(t)^2}}$$

$$\phi := d(t)\, x^2 + b(t) \qquad (1)$$

Potential

```
> V := 0;
```

$$V := 0 \qquad (2)$$

Lagrangian: (I/2)(u u*_t - u_t u*) + (1/2) u_x u*_x - (1/2) |u|^4 = A^2 phi_t + (1/2)*(A_x^2+A^2 phi_x^2) -(1/2)*A^4+V*A^2

```
> LA := A^2*diff(phi,t) + (1/2)*(diff(A,x)^2+A^2*diff(phi,x)^2) -
  (1/2)*A^4 + V*A^2;
```

$$LA := B(t)^2 \left(\mathrm{e}^{-\frac{x^2}{2\, w(t)^2}}\right)^2 \left(\left(\frac{\mathrm{d}}{\mathrm{d}t}\, d(t)\right) x^2 + \frac{\mathrm{d}}{\mathrm{d}t}\, b(t)\right) + \frac{B(t)^2\, x^2 \left(\mathrm{e}^{-\frac{x^2}{2\, w(t)^2}}\right)^2}{2\, w(t)^4} + 2\, B(t)^2 \left(\mathrm{e}^{-\frac{x^2}{2\, w(t)^2}}\right)^2 d(t)^2\, x^2 - \frac{B(t)^4 \left(\mathrm{e}^{-\frac{x^2}{2\, w(t)^2}}\right)^4}{2} \qquad (3)$$

Take out t-dependence so we can compute Leff

```
> sub1 := diff(B(t),t)=Bp,diff(d(t),t)=dp,diff(w(t),t)=wp,diff(b
  (t),t)=bp:
> sub2 := b(t)=b,B(t)=B,d(t)=d,w(t)=w:
> LA2 := subs({sub1,sub2},LA);
```

$$LA2 := B^2 \left(\mathrm{e}^{-\frac{x^2}{2\, w^2}}\right)^2 \left(dp\, x^2 + bp\right) + \frac{B^2\, x^2 \left(\mathrm{e}^{-\frac{x^2}{2\, w^2}}\right)^2}{2\, w^4} + 2\, B^2 \left(\mathrm{e}^{-\frac{x^2}{2\, w^2}}\right)^2 d^2\, x^2 - \frac{B^4 \left(\mathrm{e}^{-\frac{x^2}{2\, w^2}}\right)^4}{2} \qquad (4)$$

Leff = integral of Lag, we need to assume that a>0 and xi real to be able to evaluate integrals

```
> assume(w>0);
> assume(b>0);
> assume(d>0);
```

```
> Leff := int(LA2,x=-infinity..infinity);
```

$$Leff := \frac{B^2\, dp\, w^3 \sqrt{\pi}}{2} + B^2\, bp\, w \sqrt{\pi} + \frac{B^2 \sqrt{\pi}}{4\, w} + B^2\, d^2\, w^3 \sqrt{\pi} - \frac{B^4 \sqrt{2}\, w \sqrt{\pi}}{4} \qquad \textbf{(5)}$$

Put back t-dependences so that we can do Euler-Lag

```
> subi1 := Bp=diff(B(t),t),dp=diff(d(t),t),wp=(diff(w(t),t)),bp=
  (diff(b(t),t)):
> subi2 := b=b(t),B=B(t),d=d(t),w=w(t):
```

Euler-Lagrange eqs

```
> dLwp := subs({subi1,subi2},diff(Leff,wp)):;
> dLbp := subs({subi1,subi2},diff(Leff,bp)):;
> dLdp := subs({subi1,subi2},diff(Leff,dp)):;
> dLBp := subs({subi1,subi2},diff(Leff,Bp)):;
> eq1 := diff(dLwp ,t)=subs({subi1,subi2},diff(Leff,w));
> eq2 := diff(dLbp ,t)=subs({subi1,subi2},diff(Leff,b));
> eq3 := diff(dLdp ,t)=subs({subi1,subi2},diff(Leff,d));
> eq4 := diff(dLBp ,t)=subs({subi1,subi2},diff(Leff,B));
```

$$eq1 := 0 = \frac{3\, B(t)^2 \left(\frac{d}{dt}\, d(t)\right) w(t)^2 \sqrt{\pi}}{2} + B(t)^2 \left(\frac{d}{dt}\, b(t)\right) \sqrt{\pi} - \frac{B(t)^2 \sqrt{\pi}}{4\, w(t)^2} + 3\, B(t)^2\, d(t)^2\, w(t)^2 \sqrt{\pi} - \frac{B(t)^4 \sqrt{2} \sqrt{\pi}}{4}$$

$$eq2 := 2\, B(t)\, w(t) \sqrt{\pi} \left(\frac{d}{dt}\, B(t)\right) + B(t)^2 \left(\frac{d}{dt}\, w(t)\right) \sqrt{\pi} = 0$$

$$eq3 := B(t)\, w(t)^3 \sqrt{\pi} \left(\frac{d}{dt}\, B(t)\right) + \frac{3\, B(t)^2\, w(t)^2 \sqrt{\pi} \left(\frac{d}{dt}\, w(t)\right)}{2} = 2\, B(t)^2\, d(t)\, w(t)^3 \sqrt{\pi}$$

$$eq4 := 0 = B(t) \left(\frac{d}{dt}\, d(t)\right) w(t)^3 \sqrt{\pi} + 2\, B(t) \left(\frac{d}{dt}\, b(t)\right) w(t) \sqrt{\pi} + \frac{B(t) \sqrt{\pi}}{2\, w(t)} + 2\, B(t)\, d(t)^2\, w(t)^3 \sqrt{\pi} - B(t)^3 \sqrt{2}\, w(t) \sqrt{\pi} \qquad \textbf{(6)}$$

Sove Euler-Lag ODEs simultaneously

```
> sol:=solve({eq1,eq2,eq3,eq4},{diff(B(t),t),diff(d(t),t),diff(w
  (t),t),diff(b(t),t)}):
> eqw := diff(w(t),t)=subs(sol,diff(w(t),t));
> eqb := diff(b(t),t)=subs(sol,diff(b(t),t));
> eqd := diff(d(t),t)=subs(sol,diff(d(t),t));
> eqB := diff(B(t),t)=subs(sol,diff(B(t),t));
```

$$eqw := \frac{d}{dt}\, w(t) = 2\, w(t)\, d(t)$$

$$eqb := \frac{d}{dt}\, b(t) = \frac{5\, B(t)^2 \sqrt{2}\, w(t)^2 - 4}{8\, w(t)^2}$$

$$eqd := \frac{d}{dt}\, d(t) = -\frac{8\, d(t)^2\, w(t)^4 + B(t)^2 \sqrt{2}\, w(t)^2 - 2}{4\, w(t)^4}$$

$$eqB := \frac{d}{dt}\, B(t) = -B(t)\, d(t) \tag{7}$$

>